ABSTRACT

Would you like to be more confident in producing graphs and figures? Do you understand the differences between SGPLOT, SGSCATTER and SGPANEL? Would you like to know the different layout options in Graph Template Language (GTL)? Would you like to know how to easily create industry standard graphs such as Adverse Event timelines, Kaplan-Meier plots and Waterfall plots? Finally, would you like to learn all of these methods in a relaxing open environment that fosters questions? Great, then this topic is for you. In this Hands-on SAS Graphics Session you will be skillfully guided through SGPLOT, SGSCATTER, SGPANEL and GTL. You will also complete fun and challenging SAS Graphics exercises to enable you to retain what you have learned more easily. This Hands-on SAS Graphics session is structured so that you will learn how to create the standard plots that your manager requests, how to easily create simple ad-hoc plots for your customers and also how to create complex graphics. You will be shown how to annotate your plots and this includes how to add Unicode characters on to your plots. You will find out how to create reusable templates, which can be used by your team. Years of information have been carefully condensed into this 90 minutes Hands-on SAS Graphics session. This session will give you insight and will be very interactive! Feel free to bring some of your challenging graphical questions along!

INTRODUCTION

Before giving detailed solutions on frequently asked graphic questions, an overview of the different types of SG Procedures and when to use each one is presented.

SGPLOT

Use SGPLOT when you want to produce single-cell graphs. Figure 1 below, shows an example of a shell of a single-cell graph. Figure 1 is displayed to aid in remembering the use of SGPLOT. Figure 2 is an example of a single-cell graph produced with SGPLOT and shows a box plot of the intensity by visit number for a particularly Laboratory Test.

Figure 1: SGPLOT Blank Canvas

Figure 2: SGPLOT

SGPANEL

Use SGPANEL when you want to produce multi-cell graphs. Think of SGPANEL as being similar to SGPLOT; however instead of producing single-cell graphs, you are producing multi-cell graphs.

In SGPANEL, either each row or each column must have identical values, and this can be very useful when comparing results between rows and/or columns. In this context, values are also known as tickvalues. Because each
row or each column must have identical values in SGPANEL, if you want each cell to have values that constitute its own axis range, you will need to use SGSCATTER or Graph Template Language (GTL) to achieve this. Figure 9 shows an example of how you can use GTL to enable each cell to have its own independent axes.

Figure 3 shows an example of a shell of a multi-cell graph. Typically when using SGPANEL on Study Data Tabulation Model (SDTM) data you will panel by the --TEST variables, such as LBTEST, EGTEST or VSTEST, or you will panel by the VISIT variable. In addition, when using SGPANEL on Analysis Data Model (ADaM) data, you will usually panel by the PARAM variable as in Figure 3 and Figure 4 or the AVISIT variable.

Figure 3: SGPANEL Blank Canvas

Figure 4 is an example of a multi-cell plot produced with SGPANEL and shows a box plot of the intensity by visit number paneled by the PARAM variable, which in this case are the Laboratory Tests. In Figure 4, the Phosphate intensity values are much lower than the Glucose intensity values, and because the Phosphate and Glucose cells share a common intensity axis scale, this makes the variation of Phosphate intensity values at each visit difficult to see. Figure 9 shows an example of how you can use GTL to achieve independent axes in each cell, and therefore see the variation of the Phosphate intensity values.
Figure 4: SGPANEL

![Scatter plot matrix showing parameters]

Difficult to see Phosphate intensity values on this scale.

SGCATTER

Use SGCATTER when you want to produce multi-cell scatter plots that have independent axes or matrix plots. SGCATTER has three statements that you can use to create a paneled graph of scatter plots. They are:

- PLOT
- COMPARE
- MATRIX

This paper only shows graphs from two of the statements in SCATTER: PLOT and MATRIX. Figure 5 shows an example of a shell of a multi-cell graph which has independent axes.
Figure 5: SGSCATTER Blank Canvas

Figure 6 shows an example of a multi-cell scatter plot with independent x-axis that was produced using SGSCATTER in conjunction with the PLOT statement where one of the x-axis denotes the Petal Length and the other x-axis denotes the Petal Width. With SGSCATTER it is possible to also produce a scatter plot that have both the x-axis and the y-axis independent of each other; although the caveat with this method is that the plot type needs to be a scatter plot.

Figure 7 shows an example of a multi-cell matrix plot that was produced using SGSCATTER in conjunction with the MATRIX statement. A matrix plots each variable that you specify against the other.

Figure 6: SGSCATTER – Plot

Figure 7: SGSCATTER – Matrix
GRAPHICS TEMPLATE LANGUAGE - GTL

GTL is used to produce complex graphs. An example is where you want to produce a histogram and a box plot on the same graph, such as in Figure 8. Another example of a complex graph is a graph where you want each cell to have its own independent axes and the graph is not a scatter plot, such as in Figure 9.

Figure 8: GTL – LAYOUT LATTICE, Histogram and Horizontal Box Plot

Image of a histogram and box plot.

INDEPENDENT AXES

Hebbar and Matange (2013) show how you can simulate the subsetting of observation in GTL. You can use these ideas as illustrated in Program 1 to create cells that have independent axes. Evidently, it is possible for you to create cells using GTL that have independent axes without simulating the subsetting of observations. However, using another method to create cells that have independent axes involves reformatting your dataset, so that there is an x-variable and a y-variable for every parameter that you want to plot. Therefore if you have lots of parameters, you will end up with a dataset with a large number of columns.

Program 1 shows a snippet of the full GTL code which produces the template for Figure 9. Figure 9 shows boxplots of intensity by visit number for four laboratory tests, and each laboratory test has its own independent axes. Figure 9 was created using the layouts LAYOUT LATTICE and LAYOUT OVERLAY in GTL.

In Program 1 the EVAL and IFN functions are used to only show the Creatinine intensity results in the first cell. This was then repeated for the second, third and fourth cell so that each cell only showed the result of one parameter and each cell had its own independent axes.

Regarding the IFN function, there is another function called the IFC function, and these functions often get mixed up. A way to remember the differences between the IFN and IFC functions is to remember that the “N” or “C” part refers to the type of value that the function is returnin, and does not take into consideration the value type of the logical expression. For example, for the y-axis, IFN was used to return the numeric AVAL values when the PARAM was equal to “Creatinine (umol/L)”; otherwise (numeric) missing values were returned.

Program 1

```
layout lattice / rows = 2 columns = 2;
  cell;
    cellheader;
      entry "Creatinine (umol/L)";
    endcellheader;
    layout overlay / yaxisopts = (label = "Intensity")
      xaxisopts = (label = "Visit Number");
    boxplot x = eval(ifn(param = "Creatinine (umol/L)", visitnum, .))
      y = eval(ifn(param = "Creatinine (umol/L)", aval, .)) / group = trta;
    endlayout;
  endcell;
```
Figure 9: GTL – Boxplot of Intensity by Visit Number with Independent Axes

A few drawbacks of using the method in Program 1 to obtain independent axes in each cell are:

- This is a manual method, and therefore if you have 9 cells, you will need to create 9 different cell groupings of cell header, entry, layout overlay and boxplot statements, and also carefully add each logical expression.
- The default label of the x-axis and y-axis will need to be changed, otherwise a string that looks very similar to the expression will be shown as the x-axis and y-axis label on each cell.

To address a few of the drawbacks in Program 1, Program 2 was created and shows how Program 1 can be made more generalized with the aid of a Macro. In Program 2, the %DO loop is used to obtain and subsequently input the value of each parameter to subset on. The following code, which is extracted from Program 2, uses %QSCAN to obtain the value of the parameter based on the value of the macro variable i.

```
%LET parameter = %QSCAN(&index, &i, %str(,));
```
Program 2

```sas
%macro plot(index);

proc template;
  define statgraph independentaxislayoutmac;
  begingraph;
    layout lattice / rows = 2 columns = 2;

    %do i = 1 %to 4;
      %let parameter = %qscan(&index, &i, %str(,));
    %let parameter = %qscan(&index, &i, %str(,));
    cell;
      cellheader;
      entry "&parameter.";
    endcellheader;
    layout overlay / yaxisopts = (label = "Intensity")
                     xaxisopts = (label = "Visit Number");
    boxplot x = eval(ifn(param = "&parameter", visitnum, .))
                 y = eval(ifn(param = "&parameter", aval, .)) / group = trta;
    endlayout;
    endcell;
  %end;
endlayout;
endgraph;
%mend;

%plot(%str(Creatinine (umol/L),Gamma Glutamyl Transferase (U/L),Glucose (mmol/L),Phosphate (mmol/L)));
```

Program 2 can be further generalized by using a macro variable that contains the parameters you want to plot instead of manually entering the parameters themselves.

Program 3 shows this further generalization using the INTO clause to create two macro variables. The first macro variable `count_param` contains the number of parameters, and the second macro variable `parameterlist` contains the list of parameters (separated by a comma) that we wish to plot. The CATS function is needed to ensure that the `parameterlist` is being interpreted as a string and therefore `parameterliststring` was created. Program 3 can be further generalized by substituting `count_param` for the number 4 in the %DO loop and using macro variables to automatically specify the number of rows or columns of cells that you wish to use in your output.

Program 3

```sas
proc sql;
  select distinct count(distinct param), param into :count_param, :parameterlist
  separated by "," from adlbc4;
quit;

%let parameterliststring = %sysfunc(cats((&parameterlist))); %plot(%str(&parameterliststring));
```
KAPLAN-MEIER

Firstly, Kuhfeld and So (2014) show numerous examples on how to create and customize the Kaplan-Meier survival curve in PROC LIFETEST, and it is helpful to read this paper. An alternative method to create Kaplan-Meier survival curves is to output the statistics and estimates from LIFETEST into a dataset and then subsequently use GTL code to produce a survival curve. The GTL code can be constructed to output survival curves, which include a table of the median survival times, the number of events, and the hazard ratio etc.

This paper, builds upon the Harris (2013) paper. Harris (2013) shows extracts of the GTL code that can be used to produce a survival curve. Although, there are a few things you need to be aware of when using the datasets that are outputted by LIFETEST as your source data for your survival curves. One of the things you need to be aware of is that if you specify the ATRISK option in PROC LIFETEST, and you output the PRODUCTLIMITESTIMATES table such as in Program 4, then in your PRODUCTLIMITESTIMATES dataset, there will be a column named NUMBERATRISK, such as in Table 1. I know it is confusing, but you should use the column named LEFT instead of the NUMBERATRISK column when you want to display the number of subjects at risk on your survival curve.

Figure 10 is an example of a survival curve the PROC LIFETEST produces. Figure 11 is an example of a survival curve produced using GTL, using the datasets outputted by PROC LIFETEST. The GTL code in Figure 11, uses the NUMBERATRISK column, instead of the LEFT column. If you compare the at risk numbers in Figure 10 with Figure 11, you will see that there are differences in the last two columns of at risk numbers. For example, at 450 days, Figure 10 shows that there are no subjects at risk in the Treatment 2 group; however Figure 11 shows that at 450 days, there is one subject at risk in the Treatment 2 group. The subjects at risk in Figure 10 have been confirmed as correct, and therefore if you use the data in the PRODUCTLIMITESTIMATES dataset to plot the subjects at risk, then you should use the LEFT column, such as in Figure 12.

Program 4: LIFETEST Code

```sas
ods output Quartiles=Quartiles(where=(Percent=50));
ods output CensoredSummary=CensoredSummary(where=(control_var ne "-"));
ods output ProductLimitEstimates =ProductLimitEstimates;
ods output HomTests=HomTests(where=(test="Log-Rank"));
proc lifetest data = adtte outsurv = outsurv atrisk timelist = 0 to 540 by 90
plots=survival(atrisk=0 to 540 by 90);
    time AVAL * CNSR(1);
    strata TRTPN;
run;
```

Table 1 shows the output of the PRODUCTLIMITESTIMATES. As mentioned earlier, there can be differences between the results in the LEFT column (labeled Number Left) and the NUMBERATRISK column (labeled Number at Risk), and you should use the LEFT column if you want to show the number of subjects at risk.

Table 1: Product Limit Estimates

<table>
<thead>
<tr>
<th>STRATUM</th>
<th>TRTPN</th>
<th>TIMELIST Value</th>
<th>surv</th>
<th>Censoring Indicator</th>
<th>Survival</th>
<th>Failure</th>
<th>Survival Standard Error</th>
<th>Number Failed</th>
<th>Number Left</th>
<th>Number at Risk</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1</td>
<td>0.000</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>100</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>90.000</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>100</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>180.000</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>100</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>270.000</td>
<td>0</td>
<td>0.8540</td>
<td>0.1457</td>
<td>0.0050</td>
<td>0</td>
<td>0</td>
<td>78</td>
<td>78</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>360.000</td>
<td>0</td>
<td>0.0097</td>
<td>0.7933</td>
<td>0.0490</td>
<td>0.0257</td>
<td>71</td>
<td>14</td>
<td>14</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>450.000</td>
<td>415.673</td>
<td>0</td>
<td>0.5520</td>
<td>0.5490</td>
<td>0.0257</td>
<td>71</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td>1</td>
<td>1</td>
<td>540.000</td>
<td>475.498</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>0.000</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>100</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>90.000</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>100</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>180.000</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>100</td>
<td>100</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>270.000</td>
<td>0</td>
<td>0.8540</td>
<td>0.1457</td>
<td>0.0050</td>
<td>0</td>
<td>0</td>
<td>78</td>
<td>78</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>360.000</td>
<td>0</td>
<td>0.0097</td>
<td>0.7933</td>
<td>0.0490</td>
<td>0.0257</td>
<td>71</td>
<td>14</td>
<td>14</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>450.000</td>
<td>415.673</td>
<td>0</td>
<td>0.5520</td>
<td>0.5490</td>
<td>0.0257</td>
<td>71</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>540.000</td>
<td>475.498</td>
<td>0</td>
<td>1.0000</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>1</td>
</tr>
</tbody>
</table>
```

Difference between LEFT and NUMBERATRISK column. Use LEFT column.
Table 2 is created using the OUTSURV option in PROC LIFETEST and contains the survival estimates and confidence limits. When the maximum value within a stratum is censored, then the SURVIVAL value (Survival Distribution Function Estimate) is missing for the corresponding maximum value, and occasionally is missing at previous time-points. The issue with the survival time being missing at the higher time-points is that when using the OUTSURV dataset to plot the survival curve, the curves do not look like the survival curves that are produced by the PROC LIFETEST which is incorrect as results are not present for the maximum time in the OUTSURV dataset which will lead to the curves being incorrect as at least one curve will not show the last subjects’ censored result; These are displayed in Figure 11 and Figure 12. If you look at Figure 10, you will notice that there is a subject in Treatment Group 2, which was censored after day 400. However, Figure 11 and Figure 12 do not show this.

A solution for when the survival distribution function estimate is not present at the higher time-points, and should be present, is to calculate the minimum survival distribution function estimate of each stratum and then merge the calculated minimum survival distribution function estimate onto every result that has a missing survival distribution function estimate. The merge should be done at the stratum level. Figure 13 displays the corrected curve and mirrors Figure 10.

**Table 2: Outsurv**

<table>
<thead>
<tr>
<th>strm</th>
<th>sales</th>
<th>Censoring</th>
<th>Surv Dist Func Est</th>
<th>SDF Lower 95%</th>
<th>SDF Upper 95%</th>
</tr>
</thead>
<tbody>
<tr>
<td>189</td>
<td>2</td>
<td>313.6672586</td>
<td>0.1438322368</td>
<td>0.082731707</td>
<td>0.2211770815</td>
</tr>
<tr>
<td>190</td>
<td>2</td>
<td>314.4869284</td>
<td>0.1327682186</td>
<td>0.0742041045</td>
<td>0.2086565336</td>
</tr>
<tr>
<td>191</td>
<td>2</td>
<td>315.4955987</td>
<td>0.1217024004</td>
<td>0.065852023</td>
<td>0.1958032262</td>
</tr>
<tr>
<td>192</td>
<td>2</td>
<td>321.6498563</td>
<td>0.1108401822</td>
<td>0.057598368</td>
<td>0.1328742334</td>
</tr>
<tr>
<td>193</td>
<td>2</td>
<td>322.837236</td>
<td>0.099576164</td>
<td>0.0497483177</td>
<td>0.16397077</td>
</tr>
<tr>
<td>194</td>
<td>2</td>
<td>327.7116744</td>
<td>0.0895121457</td>
<td>0.0420345717</td>
<td>0.156436384</td>
</tr>
<tr>
<td>195</td>
<td>2</td>
<td>340.9552579</td>
<td>1.0885121457</td>
<td></td>
<td></td>
</tr>
<tr>
<td>196</td>
<td>2</td>
<td>340.5013565</td>
<td>0.0750675535</td>
<td>0.032315741</td>
<td>0.141764368</td>
</tr>
<tr>
<td>197</td>
<td>2</td>
<td>341.7017348</td>
<td>0.0632229612</td>
<td>0.0249765084</td>
<td>0.1266764887</td>
</tr>
<tr>
<td>198</td>
<td>2</td>
<td>360.7928617 0.050578369</td>
<td>0.0173896518</td>
<td>0.1110356328</td>
<td></td>
</tr>
<tr>
<td>199</td>
<td>2</td>
<td>366.60551958</td>
<td>0.0379337767</td>
<td>0.0106420823</td>
<td>0.039471691</td>
</tr>
<tr>
<td>200</td>
<td>2</td>
<td>368.54317083</td>
<td>0.0252891845</td>
<td>0.005027215</td>
<td>0.0776954516</td>
</tr>
<tr>
<td>201</td>
<td>2</td>
<td>410.32111069</td>
<td>0.0126445922</td>
<td>0.0011220265</td>
<td>0.0500736046</td>
</tr>
<tr>
<td>202</td>
<td>2</td>
<td>412.57200586</td>
<td>1</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

**Figure 10:** Survival Curve using LIFETEST

**Figure 11:** Survival Curve using GTL and NUMBERATRISK column
CONCLUSION
SGPLOT, SG PANEL, and SG SCATTER are very useful for quickly plotting basic single-cell and multi-cell figures; for more advanced figures GTL is required.

The majority of plots from the very simple to the very complex can be produced by SGPLOT, SG PANEL, LAYOUT LATTICE and LAYOUT OVERLAY.

More examples will be available during the hands-on training session, such as displaying Unicode symbols and the advantages and disadvantages of using Attribute Maps. So please attend the Picture this: Hands-on SAS Graphics Session!

REFERENCES

ACKNOWLEDGMENTS
I would like to thank Adrienne Bonwick for reviewing this paper. I would also like to thank Sharon Carroll for the support she gave me whilst preparing the paper.

RECOMMENDED READING

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:

Name: Kriss Harris
Enterprise: SAS Specialists Ltd.
%let outpath =U:\pharmasug\2015\ZQJ_HOWHarris\HT01\output\figures\other plots;
%let progname=kaplan;

data adtte;
   call streaminit(1000888);
   do i = 1 to 100;
      aval = rand("normal",310,50);
      trtpn = 1;
      cnsr = int(round(rand("uniform")+0.2,0.1));
      output;
   end;
   do i = 101 to 200;
      aval = rand("normal",250,50);
      trtpn = 2;
      cnsr = int(round(rand("uniform"),0.1));
      output;
   end;
run;

proc sort data = adtte;
   by trtpn aval;
run;

data adtte;
   set adtte;
   by trtpn aval;
   if trtpn = 2 and last.trtpn then cnsr = 1;
run;

/* Finding out the maximum for calculating at risk intervals and plotting the graph */
%let interval = 90;

proc sql;
   select ceil(max(aval)/&interval.)*&interval. into: max
   from adtte;
quit;

ods graphics / reset = all;
ods output Quartiles=Quartiles(where=(Percent=50));
ods output CensoredSummary=CensoredSummary(where=(control_var ne "-"));
ods output ProductLimitEstimates =ProductLimitEstimates;
ods output HomTests=HomTests(where=(test="Log-Rank"));
proc lifetest data = adtte outsurv = outsurv atrisk timelist = 0 to &max. by &interval. plots=survival(atrisk=0 to &max. by &interval.);
   time AVAL * CNSR(1);
   strata TRTPN;
run;

data quartiles2;
set quartiles;
LowerLimVal=strip(put(LowerLimit, 5.1));
if LowerLimVal='' then LowerLimVal=' NE';
UpperLimVal=strip(put(UpperLimit, 5.1));
if UpperLimVal='' then UpperLimVal=' NE';
mediantext = cat(strip(put(estimate, 5.1)), " (", LowerLimVal, " - ", UpperLimVal, ")");
keep STRATUM mediantext;
run;

data Censoredsummary2;
  set Censoredsummary;
  PctFailed = cats("","put(((Failed/Total) * 100),8.1), ")");
  PctCensored = cats("","put(((Censored/Total) * 100),8.1), ")");
  Total_text = strip(put(Total, best.));
  Event_text = catx(" ", put(Failed, best.), PctFailed);
  Censored_text = catx(" ", put(Censored, best.), PctCensored);
  keep stratum Total_text Event_text Censored_text;
run;

/* Merging censored summary with quartiles, to obtain final (bottom) table on graphic */
proc sql;
  create table censored_quartile as
  select a.*, b.mediantext
  from Censoredsummary2 as a
  inner join Quartiles2 as b
  on a.stratum = b.stratum;
quit;

data atrisk;
  set Productlimitestimates;
  Atrisk = strip(put(left, best.));
  rename STRATUM = classatrisk;
  rename Timelist = Tatrisk;
  keep STRATUM Timelist atrisk;
run;

/* Deriving minimum survival value and merging it back, so that KM plot goes horizontally until the end. Previously it ended quite prematurely. */
proc sql;
  create table minimum_survival_value as
  select trtpn, min(SURVIVAL) as min_survival
  from outsurv
  where survival ne .
  group by trtpn;
quit;

data outsurv2;
  merge outsurv minimum_survival_value;
  by trtpn;
  if survival = . then survival = min_survival;
run;

/* Creating final dataset */
data outsurvfinal;
  merge outsurv2 atrisk censored_quartile(rename=( Stratum = stratum_table));
  if _CENSOR_ = 1 then CENSOR_SURVIVAL = SURVIVAL;
run;

/* Creating Macro Variables to display table of events and censoring etc. */
proc sql;
select Total_text, Event_text, Censored_text, mediantext into:Total_text1 - :Total_text2, :Event_text1 - :Event_text2, :Censored_text1 - :Censored_text2, :mediantext1 - :mediantext2 from censored_quartile;
quit;

/* Creating Macro Variables to display log rank p-value */
proc sql;
select put(ProbChiSq, PVALUE6.4) into: log_rank_pvalue from homtests;
quit;

proc template;
define statgraph kmtemplate;
nmvar max interval;
mvar Total_text1 Event_text1 Censored_text1 Total_text2 Event_text2 Censored_text2 Mediantext1 Mediantext2 log_rank_pvalue;
begingraph / designwidth=896 designheight=672 border = false;
  layout lattice / rows = 3 rowweights = (0.71 0.13 0.16) rowgutter = 15
columndatarange=unionall;
  layout overlay / xaxisopts = (label = "Survival Time (Days)" linearopts =(viewmin = 0 viewmax = max tickvaluesequence = (start = 0 end = max increment = interval)))
      yaxisopts = (label = "Survival Probability")
      walldisplay = all;
stepplot x = aval y = survival / group = stratum name="Survival"
  legendlabel="Survival" lineattrs = (thickness = 2px);
scatterplot x = aval y = censor_survival / group = stratum markerattrs =
  (symbol = plus size = 10px);
  discretelegend "Survival" / location = inside across = 1 down = 3 autoalign =
  (topright);
entry "Logrank: p = " log_rank_pvalue / autoalign = (bottomleft);
endlayout;

cell;
cellheader;
  entry halign = left "Subjects at Risk:";
endcellheader;
layout overlay / xaxisopts = (display = none linearopts =
      (viewmin = 0 viewmax =
       max tickvaluesequence = (start = 0 end = max increment = interval)))
      yaxisopts = (display = (tickvalues) type =
       discrete reverse=true) border = false;
scatterplot x = Tatrisk y=eval(ifc(classatrisk = 1, "Treatment 1", "") /
  markercharacter = eval(ifc(classatrisk = 1, atrisk, ""));
scatterplot x = Tatrisk y=eval(ifc(classatrisk = 2, "Treatment 2", "") /
  markercharacter = eval(ifc(classatrisk = 2, atrisk, ""));
endlayout;
endcell;

layout overlay / xaxisopts=(display=none);
layout gridded / columns=5 rows = 3 border = true;
  entry "Arm"; entry "No. of Subjects"; entry "Events"; entry "Censored";
  entry "Median Survival (95% C.I.)";
  entry "Treatment 1"; entry Total_text1; entry Event_text1; entry Censored_text1;
  entry "Treatment 2"; entry Total_text2; entry Event_text2; entry Censored_text2;
  entry Mediantext1; entry Mediantext2;
endlayout;
endlayout;
endlayout;
endgraph;
end;
run;

proc format;
value trt 1 = "Treatment 1"
2 = "Treatment 2";
run;

proc template;
define style styles.KJHPlot;
parent = Styles.listing;
style GraphFonts from GraphFonts
   "Fonts used in graph styles"
   / 'GraphDataFont' = ("<sans-serif>, <MTsans-serif>" ,8pt)
   'GraphValueFont' = ("<sans-serif>, <MTsans-serif>" ,8pt)
   'GraphLabelFont' = ("<sans-serif>, <MTsans-serif>" ,8pt);
CLASS graphWalls / FRAMEBORDER=off;
end;
run;

ods graphics / reset = all height = 3in width = 4.5in imagename = "sprogname."
ods listing gpath = sopath . image_dpi = 200 style = styles.KJHPlot;
proc sgrender data = outsurvfinal template = kmtemplate;
format CLASSATRISK stratum trt .
run;
ods graphics / reset = all;